
London summit – oder „was ist neu in Excel?“ 

Excelstammtisch vom 10.03.2025 

1. Der London summit 

1.1. Themen 

1.2. Zusammenfassung von Leila Gharani: Was ist neu? 

1.2.1. Fokuszelle 

 

 

1.2.2. Doppelpunkt Punkt  

A1:A1 

A1:A17 

A1:G17 

$A$1:$G$17 

A1.:A17 

A1:.A17 

A1.:.A17 

ABSCHNBEREICH 

=SORTIEREN(A2:D11;3) 

Abbildung 1.1  SORTIEREN „schießt“ übers Ziel hinaus. 

Möglicherweise vergrößert sich die Quellliste. Deshalb wurde „übers Ziel rausgeschossen“ und 

nun stehen störende Nullen am Ende der Zielliste. Die Lösung heißt „.“: 

=SORTIEREN(A2:.D11;3) 

Abbildung 1.2  SORTIEREN jetzt mit Punkt. 

Abbildung 1.3  SORTIEREN jetzt mit Punkt. 

=$A$2:.$A$25 

Abbildung 1.4  Der Punkt in der Datenüberprüfung 

Abbildung 1.5  Die Liste wird dynamisch erweitert. 

=ZÄHLENWENN(Tabelle1!$B$3:.$B$48;$A6)>0 

Abbildung 1.6  Die bedingte Formatierung. 

Abbildung 1.7  Weitere Einträge folgen 

Abbildung 1.8  Das Ergebnis der aktualisierten bedingten Formatierung 

Abbildung 1.9  Schade – auch in Diagrammen wäre der Punkt nützlich … 

1.2.3. Die regulären Ausdrücke 



London summit – oder „was ist neu in Excel?“ // Excelstammtisch vom 10.03.2025 

5 

Abbildung 1.10  RegEx in VBA 

document.getElementById('emailForm').addEventListener  

('submit', function(event) { 

    event.preventDefault(); 

 

    const emailInput =  

        document.getElementById('email').value; 

    const result = document.getElementById('result'); 

     

    // Regulärer Ausdruck für die E-Mail-Überprüfung 

    const emailPattern = /^[a-zA-Z0-9._-]+@[a-zA-Z0-9.-]+\.[a-zA-

Z]{2,6}$/; 

 

    if (emailPattern.test(emailInput)) { 

        result.textContent =  

           'Die E-Mail-Adresse ist gültig.'; 

        result.style.color = 'green'; 

    } else { 

        result.textContent =  

           'Die E-Mail-Adresse ist ungültig.'; 

        result.style.color = 'red'; 

    } 

}); 

Abbildung 1.11  RegEx in JavaScript 

Tabelle 1.1  Liste der Zeichen von RegEx 

Zeichen Wirkung/Einsatz 

Beliebiges 
Zeichen 

Steht für das Zeichen, falls nicht anders angegeben. 

. Steht für ein beliebiges einzelnes Zeichen außer einem Zeilen- oder einem 
Absatzumbruch. Beispielsweise liefert der Suchbegriff "Schmi.t" sowohl 
"Schmitt" als auch "Schmidt". 

^ Findet den Suchbegriff nur, wenn er am Absatzanfang steht. Sonderinhalte 
wie beispielsweise Leerfelder und an Zeichen verankerte Rahmen am 
Absatzanfang werden ignoriert. Beispiel: "^Peter". 

$ Findet den Suchbegriff nur, wenn er am Absatzende steht. Sonderinhalte wie 
beispielsweise Leerfelder und an Zeichen verankerte Rahmen am Absatzende 
werden ignoriert. Beispiel: "Peter$". 

$ allein stimmt mit dem Ende eines Absatzes überein. So ist es möglich, 
Absatzumbrüche zu suchen und zu ersetzen. 

* Findet keines oder mehr der Zeichen vor dem "*". So liefert etwa der 
Suchbegriff "Ab*c" die Einträge "Ac", "Abc", "Abbc", "Abbbc" und so weiter. 

+ Findet ein oder mehr der Zeichen vor dem "+". Beispielsweise findet "AX.+4" 
zwar "AXx4", jedoch nicht "AX4". 

Es wird immer die längstmögliche Zeichenkette gefunden, die dem 
Suchmuster in einem Absatz entspricht. Wenn der Absatz die Zeichenkette "AX 
4 AX4" enthält, wird der gesamte Ausdruck hervorgehoben. 

? Findet keines oder eines der Zeichen vor dem "?". Beispielsweise findet 
"Texts?" "Text" und "Texts" und "x(ab|c)?y" findet "xy", "xaby" oder "xcy". 

\ Die Suchfunktion interpretiert das Sonderzeichen nach dem "\" als ein 
normales Zeichen und nicht als einen regulären Ausdruck (außer bei den 
Kombinationen \n, \t, \> und \<). Beispielsweise wird bei der Suche nach 
"kein\." zwar "kein." gefunden, jedoch nicht "keine" oder "keins". 

\n Steht für einen mit [Umschalt] + [Eingabetaste] eingefügten Zeilenumbruch. 
Sie tauschen Zeilenumbrüche gegen Absatzumbrüche aus, indem Sie \n in 
die Felder Suchen und Ersetzen eingeben und dann einen Suchen-und-
Ersetzen-Vorgang ausführen. 

\n steht im Textfeld Suchen für einen Zeilenumbruch, der mit der 
Tastenkombination [Umschalt] + [Eingabetaste] eingegeben wurde. 

\n steht im Textfeld Ersetzen für einen mit der Eingabetaste oder Enter 
eingegebenen Absatzumbruch. 

\t Steht für ein Tabulatorzeichen. Dieser Ausdruck kann auch im Feld Ersetzen 
verwendet werden. 

\b Findet eine Wortgrenze. Zum Beispiel findet "\bbuch" das Wort "Buchstabe" 
aber nicht "Textbuch", wohingegen "buch\b" das Wort "Textbuch" findet aber 
nicht "Buchstabe". 

^$ Findet einen leeren Absatz. 

^. Findet das erste Zeichen eines Absatzes. 

& oder $0 Fügt die Zeichenkette, die durch die Suchkriterien im Textfeld Suchen 
gefunden wurde, zu dem Begriff im Feld Ersetzen hinzu, wenn Sie die 
Ersetzung durchführen. 

Wenn Sie im Textfeld Suchen zum Beispiel "Fenster" und im Textfeld Ersetzen 
"&rahmen" eingeben, wird das Wort "Fenster" durch "Fensterrahmen" ersetzt. 

Außerdem können Sie mit dem Ausdruck "&" im Textfeld Ersetzen die Attribute 
oder das Format des durch die Suchkriterien gefundenen Suchbegriffs 
ändern. 

[abc123] Steht für eines der Zeichen in der Klammer. 

[a-e] Steht für irgendein Zeichen zwischen a und e, einschließlich beider Start- und 
Endzeichen. 

Zeichen Wirkung/Einsatz 

Die Zeichen werden anhand ihrer Codenummer angeordnet. 

[a-eh-x] Steht für ein beliebiges Zeichen im Buchstabenbereich a-e und h-x. 

[^a-s] Steht für ein beliebiges Zeichen, das sich nicht im Buchstabenbereich 
zwischen a und s befindet. 

\uXXXX Steht für ein Zeichen auf Grundlage seines vierstelligen Hexadezimal-
Unicodes (XXXX). 

\UXXXXXXXX Für unbedeutende Zeichen gibt es eine separate Variante mit groß 
geschriebenem U und acht Hexadezimalzahlen (XXXXXXXX). 

  Der Code für einige Zeichens hängt von der jeweiligen Schrift ab. Die Codes 
können Sie unter Einfügen - Symbol einsehen. 

| Findet die Begriffe, die vor dem "|" vorkommen und auch die, die nach dem 
"|" auftauchen. "dies|das" findet "dies" und "das". 

{2} Gibt an, wie oft das Zeichen vor der öffnenden Klammer vorkommen muss. 
Zum Beispiel liefert der Suchbegriff "Man{2}" das Wort "Mann". 

{1,2} Gibt sowohl die minimale als auch maximale Anzahl an, die das Zeichen in 
vor der sich öffnenden Klammer vorkommen kann. "Man{1,2}" findet und 
wählt sowohl "Man" als auch "Mann". 

{1,} Gibt an, wie oft das Zeichen vor der öffnenden Klammer im Wort mindestens 
vorkommen muss. Beispiel: Der Suchbegriff "Man{2,}" findet "Mann", 
"Mannn" und "Mannnn". 

( ) Im Textfeld Suchen: 

Die in der Klammer enthaltenen Zeichen gelten als Referenz. Auf die erste 
Referenz im aktuellen Ausdruck können Sie dann mit "\1" Bezug nehmen, auf 
die zweite mit "\2" und so weiter. 

Enthält Ihr Text zum Beispiel die Zahl 13487889 und Sie führen eine Suche 
mit dem Ausdruck (8)7\1\1 durch, dann wird "8788" gefunden. 

Zum Gruppieren von Suchtext kann () verwendet werden, "a(bc)?d" findet 
beispielsweise "ad" oder "abcd". 

Im Textfeld Ersetzen: 

Um Referenzen zu ersetzen, verwenden Sie statt \ (umgekehrter Schrägstrich) 
$ (Dollar). Um die gesamte gefundene Zeichenkette zu ersetzen, verwenden 
Sie $0. 

REGEXTESTEN 

M(a|e)(i|y)e?r 

=REGEXTESTEN(D2:D300001;"M(a|e)(i|y)e?r") 

Abbildung 1.12  REGEXTESTEN liefert WAHR oder FALSCH 

=FILTER(D1:D300001;E1:E300001=WAHR) 

=REGEXTESTEN(B2:B51;"[0-9]{3}[A-Z]{3}[0-9]{6}[A-Z]") 

Abbildung 1.13  REGEXTESTEN liefert WAHR oder FALSCH 

=FILTER(B5:B459;REGEXTESTEN(B5:B459;".*"& 

TEXTVERKETTEN("|";WAHR;tblListe[Liste]&".*"))) 

Abbildung 1.14  Mit REGEXTESTEN können bestimmte Zeilen gefiltert werden. 

Und so kann man auch die Elemente aufnehmen, die gefiltert wurden: 

=FILTER(B5:B459&" enthält: "& 

REGEXEXTRAHIEREN(B5:B459; 

TEXTVERKETTEN("|";WAHR;tblListe[Liste]);1); 

REGEXTESTEN(B5:B459; 

TEXTVERKETTEN("|";WAHR;tblListe[Liste]))) 

=LET(TK;TEXTVERKETTEN("|";WAHR;tblListe[Liste]); 

FILTER(B5:B459&" enthält: "& 

REGEXEXTRAHIEREN(B5:B459;TK;1);REGEXTESTEN(B5:B459;TK))) 

Abbildung 1.15  Mit REGEXTESTEN können bestimmte Zeilen gefiltert werden. 

REGEXEXTRAHIEREN 

=TEXTKETTE(REGEXEXTRAHIEREN(B2;"[0-9]+";1)) 

Abbildung 1.16  Mit REGEXEXTRAHIEREN können Ziffern ausgelesen werden. 

Tabelle 1.2  der return_mode bei REGEXEXTRAHIEREN 

Return_mode  Beschreibung 

0 Erste Übereinstimmung (Standard) 

Return_mode  Beschreibung 

1 Alle Übereinstimmungen 

2 Erfassen von Gruppen der ersten Übereinstimmung 

=TEXTKETTE(REGEXEXTRAHIEREN(B2;"\d+";1)) 

=TEXTKETTE(REGEXEXTRAHIEREN(B2;"[A-Z]+";1)) 

=TEXTKETTE(REGEXEXTRAHIEREN(B2;"\D+";1)) 

=REGEXEXTRAHIEREN(C2:C55;"[A-Z][\p{Ll}]+") 

=REGEXEXTRAHIEREN(C2:C111;"[A-Z][\p{Ll}]+$") 

Abbildung 1.17  Mit REGEXEXTRAHIEREN können Texte ausgelesen werden. 

REGEXERSETZEN 

→

=REGEXERSETZEN(G6;"[^\d]";"") 

=REGEXERSETZEN( 

 H6; 

  "(^\d{6})(\d{5})(\d{1})$"; 

  "DE $1 $2\.$3" 

) 

Abbildung 1.18  Mit REGEXERSETZEN können Texte transformiert werden. 

Tabelle 1.3  Zusammenfassung: RegEx-Textfunktionen 

Funktionsname  Beschreibung Beispiel 

REGEXTESTEN Prüft, ob eine Text mit dem 
Muster übereinstimmt 

=REGEXTESTEN(D2; 

"M(a|e)(i|y)e?r") 

REGEXEXTRAHIEREN Extrahiert Teilstrings aus 
einem Text 

=REGEXEXTRAHIEREN(B2; 

"[0-9]+";1) 

REGEXERSETZEN Gibt einen Text gemäß dem 
Muster zurück 

=REGEXERSETZEN(G6; 

"[^\d]";"") 

 

1.3. GRUPPIERENNACH und PIVOTMIT (Vortrag von Alan Murray) 

=GRUPPIERENNACH(B1:B76;F1:F76;SUMME) 



London summit – oder „was ist neu in Excel?“ // Excelstammtisch vom 10.03.2025 

12 

 

Abbildung 1.19  GRUPPIERENNACH – Variante I 

=GRUPPIERENNACH(tbl_Verkäufe[Verkäufer]; 

tbl_Verkäufe[Umsatz];SUMME) 

Neben SUMME stehen die weiteren statistischen Funktionen zur Verfügung: 

Abbildung 1.20  GRUPPIERENNACH – Der Parameter function 

Hinweis 

Abbildung 1.21  Für die Funktion wird ein Fehler gemeldet – es handelt sich aber nicht um einen Fehler 

Abbildung 1.22  GRUPPIERENNACH und PROZENTVON 

=GRUPPIERENNACH(tbl_Artikel[Kategorie-Nr]; 

tbl_Artikel[Artikelname];TEXTKETTE) 

Abbildung 1.23  GRUPPIERENNACH und TEXTKETTE 

=WECHSELN(GRUPPIERENNACH(tbl_Artikel[Kategorie-Nr]; 

tbl_Artikel[Artikelname];MATRIXZUTEXT;0);"; ";ZEICHEN(10)) 

Abbildung 1.24  GRUPPIERENNACH und MATRIXZUTEXT 

Abbildung 1.25  Der Parameter field_headers von GRUPPIERENNACH 

Abbildung 1.26  Der Parameter total_depth von GRUPPIERENNACH 

Tabelle 1.4  Der Parameter total_depth bei GRUPPIERENNACH: 

Wert Beschreibung Beispiel:  

0 Keine Summen 

 

1 Gesamtergebnisse 

 

2 Gesamt- und Teilergebnisse 

 

-1 Gesamtsummen oben 

 

 Gesamt- und Teilergebnisse oben 

 

Hinweis 

=GRUPPIERENNACH(tbl_Verkäufe[[#Alle]; 

[Verkäufer]:[Artikel]]; … 

=GRUPPIERENNACH(A2:B46; … 

HSTAPELN(tbl_Verkäufe[[#Alle]; 

[Verkäufer]];tbl_Verkäufe[[#Alle];[Kunde]]) 

HSTAPELN(A2:A46;C2:C46) 

=GRUPPIERENNACH( 

HSTAPELN(tbl_Verkäufe[[#Alle];[Verkäufer]]; 

tbl_Verkäufe[[#Alle];[Kunde]]); 

tbl_Verkäufe[[#Alle];[Umsatz]]; 

SUMME;3;-2) 

=GRUPPIERENNACH(tbl_Verkäufe[[#Alle]; 

[Verkäufer]:[Artikel]]; 

tbl_Verkäufe[[#Alle];[Umsatz]]; 

SUMME;3;1;-3) 

Abbildung 1.27  Der Parameter sort_order von GRUPPIERENNACH 

=GRUPPIERENNACH(tbl_Verkäufe[[#Alle]; 

[Verkäufer]:[Artikel]]; 

tbl_Verkäufe[[#Alle];[Umsatz]]; 

SUMME;3;1;-3; 

tbl_Verkäufe[[#Alle];[Artikel]]<>"Briefpapier") 

… 

tbl_Verkäufe[[#Alle];[Menge]]>=50) 

=GRUPPIERENNACH( 

tbl_Verkäufe[[#Alle];[Verkäufer]:[Artikel]]; 

tbl_Verkäufe[[#Alle];[Umsatz]]; 

SUMME;3;1;-3; 

tbl_Verkäufe[[#Alle];[Auswahl]]) 

Abbildung 1.28  Man kann auch Kontrollkästchen filtern 

Hinweis 

Und PIVOTMIT? 

Die Funktion PIVOTMIT arbeitet analog zu der Funktion GRUPPIERENNACH – mit dem 

Unterschied, dass sie einen Parameter mehr aufweist: Neben den row_fields gibt es auch 

col_fields: 

 

Abbildung 1.29  PIVOTMIT 

=PIVOTMIT(B1:C76;D1:D76;E1:E76;SUMME;3;1) 

1.4. Power BI in Excel (Vortrag von Chandoo) 

Chandoo hat ein Visual in Excel nachgebaut: in einer Pivottabelle gibt es zwei Measures: 

Gesamt2:= SUM('tbl_Verkäufe'[Umsatz]) 

Gesamt:= CALCULATE(sumx('tbl_Verkäufe'; 

'tbl_Verkäufe'[Umsatz]);ALL('tbl_Verkäufe'[Verkäufer])) 

Abbildung 1.30  Die ungefilterte Liste / das ungefilterte Diagramm 

Abbildung 1.31  Die gefilterte Liste / das gefilterte Diagramm 


